# $Id: PKGBUILD 175124 2013-01-14 12:47:12Z stephane $
# Maintainer: Stéphane Gaudreault <stephane@archlinux.org>
# Contributor: Allan McRae <allan@archlinux.org>
# Contributor: Andreas Radke <andyrtr@archlinux.org>

pkgname=dialog
pkgver=1.2_20121230
pkgrel=1
pkgdesc="A tool to display dialog boxes from shell scripts"
arch=('i686' 'x86_64')
url="http://invisible-island.net/dialog/"
license=('LGPL2.1')
depends=('ncurses')
source=(ftp://invisible-island.net/${pkgname}/${pkgname}-${pkgver/_/-}.tgz)
sha1sums=('f6ba16481e04c707c7cbfc3c3c1340040bdb0aa3')

build() {
  cd "${srcdir}/$pkgname-${pkgver/_/-}"
  ./configure --prefix=/usr --mandir=/usr/share/man \
              --with-ncursesw --enable-nls
  make
}

package() {
  cd "${srcdir}/$pkgname-${pkgver/_/-}"
  make DESTDIR="${pkgdir}" install install-lib
}
